
Ada Database Objects Programmer's Guide

Stephane Carrez

2025-07-05

Contents

1	Introduction	3
2	Installation	4
2.1	Using Alire	4
2.2	Without Alire	4
2.2.1	Database Driver Installation	4
2.2.2	Before Building	5
2.2.3	Configuration	6
2.2.4	Build	6
2.2.5	Installation	7
2.3	Using	7
3	Tutorial	8
3.1	Getting a Database Connection	8
3.2	Opening a Session	9
3.3	Inserting a database record	9
3.4	Executing SQL queries	10
3.5	Getting a list of objects	11
3.6	Running SQL queries	11
4	ORM Tutorial	12
4.1	Defining the data model	12
4.2	Generating the Ada model and SQL schema	14
4.3	Getting a Database Connection	15
4.4	Opening a Session	16
4.5	Creating a database record	16
4.6	Loading a database record	17
4.7	Getting a list of objects	17
4.8	Running SQL queries	18
5	Session	19
5.1	Connection string	19
5.2	Database Drivers	19
5.2.1	MySQL Database Driver	20
5.2.2	SQLite Database Driver	21
5.2.3	PostgreSQL Database Driver	22
5.3	Session Factory	22

5.4	Database Caches	23
5.5	Sequence Generators	23
5.5.1	HiLo Sequence Generator	24
5.5.2	Snowflake ID Generator	24
6	Database Statements	26
6.1	Query Parameters	26
6.1.1	Parameter Expander	27
6.2	Query Statements	27
6.3	Named Queries	27
6.3.1	XML Query File	28
6.3.2	SQL Result Mapping	28
6.3.3	SQL Queries	29
6.3.4	Using Named Queries	30
7	Model Mapping	32
7.1	Table definition	32
7.2	Column mapping	33
7.3	Primary keys	35
7.4	Relations	35
7.5	Versions	36
7.6	Loading Objects	38
7.7	Modifying Objects	39
7.8	Deleting Objects	39
8	Database schema migration	40
8.1	Migration mechanism	40
8.2	Taking into account the migration during development	42
9	Troubleshooting	43
9.1	Change the log configuration	43
9.2	Handling exceptions	44

1 Introduction

Ada Database Objects is an Ada 2012 library that provides database drivers and object relational mapping to access a database in Ada. The library supports Postgresql, MySQL/MariaDB, SQLite or SQLCIPHER as databases. Although not mandatory, it allows to map database objects into Ada records and access database content easily. Most of the concepts developed for ADO come from the Java Hibernate ORM.

The ORM uses an YAML, XML mapping file or an UML model, a code generator and a runtime library for the implementation. It provides a database driver for Postgresql, MySQL/MariaDB, SQLite or SQLCIPHER. The ORM helps your application by providing a mapping of your database tables directly in the target programming language: Ada05 in our case. The development process is the following:

- You design your database model either using a UML tool or by writing an XML or YAML description file,
- You generate the Ada05 mapping files by using the Dynamo code generator,
- You generate the SQL database tables by using the same tool,
- You write your application on top of the generated code that gives you direct and simplified access to your database.




Figure 1: ORM Development Model

This document describes how to build the library and how you can use the different features to simplify and help you access databases from your Ada application.

2 Installation

This chapter explains how to build and install the library.

2.1 Using Alire

The Ada Database Objects library is available as several Alire crates to simplify the installation and setup your project. Run the following commands to setup your project to use the library:

```
1 alr index --update-all
2 alr with ado
```

The database drivers for SQLite/SQLCipher, MySQL/MariaDB and PostgreSQL are provided as separate crates. Choose the database drivers you want to use and run one or some of the commands below. For MariaDB use `ado_mysql`.

```
1 alr with ado_sqlite
2 alr with ado_mysql
3 alr with ado_postgresql
```

2.2 Without Alire

2.2.1 Database Driver Installation

The PostgreSQL, MySQL/MariaDB and SQLite/SQLCipher development headers and runtime are necessary for building the ADO driver. The configure script will use them to enable the ADO drivers. The configure script will fail if it does not find any database driver.

2.2.1.1 Ubuntu MySQL/MariaDB development installation

```
1 sudo apt-get install libmysqlclient-dev
```

MariaDB development installation

```
1 sudo apt-get install mariadb-client libmariadb-client-lgpl-dev
```

SQLite development installation

```
1 sudo apt-get install libsqlite3-dev
```

SQLCipher development installation

```
1 sudo apt-get install sqlcipher libsqlcipher-dev
```

PostgreSQL development installation

```
1 sudo apt-get install postgresql-client libpq-dev
```

2.2.1.2 Windows It is recommended to use msys2 available at <https://www.msys2.org/> and use the `pacman` command to install the required packages.

```
1 pacman -S git
2 pacman -S make
3 pacman -S unzip
4 pacman -S base-devel --needed
5 pacman -S mingw-w64-x86_64-sqlite3
```

For Windows, the installation is a little bit more complex and manual. You may either download the files from MySQL/MariaDB and SQLite download sites or you may use the files provided by Ada Database Objects in the `win32` directory.

For Windows 32-bit, extract the files:

```
1 cd win32 && unzip sqlite-dll-win32-x86-3290000.zip
```

For Windows 64-bit, extract the files:

```
1 cd win32 && unzip sqlite-dll-win64-x64-3290000.zip
```

If your GNAT 2019 compiler is installed in `C:/GNAT/2019`, you may install the liblzma, MySQL/MariaDB and SQLite libraries by using `msys cp` with:

```
1 cp win32/*.dll C:/GNAT/2019/bin
2 cp win32/*.dll C:/GNAT/2019/lib
3 cp win32/*.lib C:/GNAT/2019/lib
4 cp win32/*.a C:/GNAT/2019/lib
```

2.2.2 Before Building

Before building ADO, you will need:

- Ada Utility Library
- XML/Ada
- Either the PostgreSQL, MySQL/MariaDB or SQLite development headers installed.

First get, build and install the XML/Ada and then get, build and install the Ada Utility Library.

2.2.3 Configuration

The support for SQLite, SQLCipher, MySQL/MariaDB and PostgreSQL are enabled only when a `HAVE_XXX=yes` configuration variable is defined. Run the setup command that records in the `Makefile.conf` the configuration you want to build.

The `HAVE_ALIRE` configuration allows you to build with Alire or not.

The example below enables the SQLite and PostgreSQL components but disables the MySQL/MariaDB support and disables the use of Alire to build the library.

```
1 make setup BUILD=debug PREFIX=/build/install \  
2 HAVE_SQLITE=yes HAVE_SQLCIPHER=yes HAVE_POSTGRESQL=yes \  
3 HAVE_MYSQL=no HAVE_ALIRE=no
```

Note: if you plan to use both SQLite and SQLCipher databases, use the `HAVE_SQLCIPHER=yes` configuration since SQLCipher supports non-encrypted databases.

2.2.4 Build

After configuration is successful, you can build the library by running:

```
1 make
```

To use the installed libraries, make sure your `ADA_PROJECT_PATH` contains the directory where you installed the libraries (configured by the `PREFIX=<path>` option in the setup phase). The installed GNAT projects are the same as those used when using Alire.

After building, it is good practice to run the unit tests before installing the library. The unit tests are built and executed using:

```
1 make test
```

And unit tests are executed by running the `bin/ado_harness` test program. A configuration file is necessary to control the test parameters including the test database to be used. To run the tests with a MySQL/MariaDB database, use the following command:

```
1 bin/ado_harness -config test-mysql.properties
```

and with a SQLite database, use the following command:

```
1 bin/ado_harness -config test-sqlite.properties
```

2.2.5 Installation

The installation is done by running the `install` target:

```
1 make install
```

If you want to install on a specific place, you can change the `prefix` and indicate the installation direction as follows:

```
1 make install prefix=/opt
```

2.3 Using

To use the library in an Ada project, add the following line at the beginning of your GNAT project file:

```
1 with "ado";  
2 with "ado_all";
```

It is possible to use only a specific database driver, in that case your GNAT project file could be defined as follows:

```
1 with "ado";  
2 with "ado_mysql";  
3 with "ado_sqlite";  
4 with "ado_postgresql";
```

where the `ado_mysql`, `ado_sqlite` and `ado_postgresql` are optional and included according to your needs.

3 Tutorial

This small tutorial explains how an application can access a database (PostgreSQL, MySQL, SQLite or SQLCipher) to store its data by using the SQL support provided by Ada Database Objects.

The tutorial application is a simple user management database which has only one table:

```
1 CREATE TABLE IF NOT EXISTS user (  
2   `id` BIGINT UNIQUE NOT NULL,  
3   `object_version` INTEGER NOT NULL,  
4   `name` VARCHAR(256) NOT NULL,  
5   `email` VARCHAR(256) UNIQUE NOT NULL,  
6   `date` VARCHAR(256) NOT NULL,  
7   `description` VARCHAR(256) NOT NULL,  
8   `status` INTEGER NOT NULL,  
9   PRIMARY KEY (`id`)  
10 );
```

3.1 Getting a Database Connection

To access the database, we will need a database connection. These connections are obtained from a factory and they are represented by a session object.

The session factory is the entry point to obtain a database session.

```
1 with ADO.Sessions;  
2 with ADO.Sessions.Factory;  
3   ...  
4   Factory : ADO.Sessions.Factory.Session_Factory;
```

The factory can be initialized by giving a URI string that identifies the driver and the information to connect to the database. Once created, the factory returns a session object to connect to that database. To connect to another database, another factory is necessary. For SQLite and SQLCipher, the URI string must indicate the path of the database file. For MySQL and PostgreSQL, the URI string should contain the host name and port to connect to the database server.

To get access to a SQLite database, the factory could be initialized as follows:

```
1 ADO.Connections.SQLite.Initialize;  
2 ADO.Sessions.Factory.Create (Factory, "sqlite:samples.db?synchronous  
   =OFF&encoding='UTF-8'");
```

Factory initialization is done once when an application starts. The same factory object can be used by multiple tasks.

3.2 Opening a Session

The session is created by using the `Get_Session` or the `Get_Master_Session` function of the factory. Both function return a session object associated with a database connection. The `Get_Session` will return a `Session` object which is intended to provide read-only access to the database. The `Get_Master_session` returns a `Master_Session` object which provides a read-write access to the database.

In a typical MySQL or PostgreSQL Master/Slave replication, the `Master_Session` will refer to a connection to the MySQL or PostgreSQL master while the `Session` could refer to a slave. With an SQLite or SQLCipher database, both sessions will in fact share the same SQLite internal connection.

To load or save the user object in the database, we need a `Master_Session` database connection:

```
1 with ADO.Sessions;  
2 ...  
3 Session : ADO.Sessions.Master_Session := Factory.Get_Master_Session;
```

3.3 Inserting a database record

To create our first database record, we will declare a variable that will represent the new database record. The `User_Ref` represents a reference to such record.

```
1 Query : constant String := "INSERT INTO user (`id`, `object_version`,`  
2   & " `name`, `email`, `date`, `description`, `status`) VALUES(?, ?, "  
3   & "?, ?, ?, ?, ?);";  
4 Stmt   : ADO.Statements.Insert_Statement;
```

After this declaration, the variable does not refer to any database record but we can still set some fields:

```
1 DB.Begin_Transaction;  
2 Stmt := DB.Create_Statement (Query);  
3 DB.Allocate ("user", Id);  
4 Stmt.Bind_Param ("id", Id);  
5 Stmt.Bind_Param ("version", Natural (1));  
6 Stmt.Bind_Param ("name", Name);  
7 Stmt.Bind_Param ("email", Email);  
8 Stmt.Bind_Param ("date", Ada.Calendar.Clock);  
9 Stmt.Bind_Param ("description", Desc);  
10 Stmt.Bind_Param ("status", Status);  
11 Stmt.Execute (Result);  
12 DB.Commit;
```

To save the object in the database, we just need to call the `Save` operation. To save the object, we need a database session that is capable of updating and inserting new rows. If the object does not

yet have a primary key, and if the primary key allocation mode is set to `hilo`, the ADO runtime will allocate a new unique primary key before inserting the new row.

```
1 User.Save (Session);
```

The primary key can be obtained after the first `Save` with the following operation:

```
1 Id : ADO.Identifier := User.Get_Id;
```

3.4 Executing SQL queries

Loading a database record is quite easy and the ADO framework proposes two mechanisms. First, let's declare our user variable:

```
1 Stmt    : ADO.Statements.Query_Statement;  
2 Query   : constant String := "SELECT u.id, u.name, u.email FROM user AS  
    u"  
3         & " ORDER BY u.name ASC";  
4 begin  
5     Stmt := DB.Create_Statement (Query);  
6     Stmt.Execute;
```

```
1 while Stmt.Has_Elements loop  
2     Ada.Text_IO.Put (ADO.Identifier'Image (Stmt.Get_Identifier (0)));  
3     Ada.Text_IO.Set_Col (10);  
4     Ada.Text_IO.Put (Stmt.Get_String (1));  
5     Ada.Text_IO.Set_Col (60);  
6     Ada.Text_IO.Put_Line (Stmt.Get_String (2));  
7     Stmt.Next;  
8 end loop;
```

Then we can load the user record from the primary key identifier (assuming the identifier is "23"):

```
1 Harry.Load (Session, 23);
```

If the user cannot be found, the `Load` operation will raise the `NOT_FOUND` exception.

In many cases, we may not know the primary key but a search on one or several columns may be necessary. For this, we can create a query filter and use the `Find` operation. To use a query filter, we need first to declare a `Query` object:

```
1 with ADO.SQL;  
2 ...  
3 Query : ADO.SQL.Query;
```

On the query object, we have to define the filter which represents the condition and set the possible parameters used by the filter.

```
1 Query.Bind_Param (1, "Harry");
2 Query.Set_Filter ("name = ?");
```

Once the query is defined and initialized, we can find the database record:

```
1 Found : Boolean;
2 ...
3 User.Find (Session, Query, Found);
```

Unlike the `Load` operation, `Find` does not raise an exception but instead returns a boolean value telling whether the record was found or not. The database query (and filter) has to return exactly one record to consider the object as found.

3.5 Getting a list of objects

When several records have to be read, it is necessary to use the `List` operation together with a vector object.

```
1 Users : User_Vector;
```

The `List` operation gets the vector object, the database session and the query filter. If the vector contained some elements, they are removed and replaced by the query result.

```
1 List (Users, Session, Query);
```

3.6 Running SQL queries

Sometimes it is necessary to execute SQL queries to be able to get the result without having it to be mapped to an Ada record. For this, we are going to use the `ADO.Statements`

```
1 with ADO.Statements;
2 ...
3 Statement : ADO.Statements.Query_Statement := Session.
   Create_Statement ("SELECT COUNT(*) FROM user");
```

and then execute it and retrieve the result.

```
1 Statement.Execute;
2 if not Statement.Has_Elements then
3   Put_Line ("SQL count() failed")
4 else
5   Put_Line (Integer'Image (Statement.Get_Integer (0)));
6 end if;
```

4 ORM Tutorial

This small tutorial explains how an application can access a database (PostgreSQL, MySQL or SQLite) to store its data by using the Ada Database Objects framework. The framework has several similarities with the excellent Hibernate Java framework.

The ADO framework is composed of:

- A code generator provided by Dynamo,
- A core runtime,
- A set of database drivers (PostgreSQL, MySQL, SQLite).

The tutorial application is a simple user management database which has only one table.

4.1 Defining the data model

The first step is to design the data model. You have the choice with:

- Using an UML modeling tool such as ArgoUML,
- Writing an XML file following the Hibernate description,
- Writing a YAML description according to the Doctrine mapping.

In all cases, the model describes the data table as well as how the different columns are mapped to an Ada type. The model can also describe the relations between tables. XML and YAML data model files should be stored in the `db` directory.

Let's define a mapping for a simple `user` table and save it in `db/user.hbm.xml`:

```
1 <?xml version="1.0" encoding="UTF-8"?>
2 <hibernate-mapping default-cascade="none">
3   <class name="Samples.User.Model.User"
4     table="user" dynamic-insert="true" dynamic-update="true">
5     <comment>Record representing a user</comment>
6     <id name="id" type="ADO.Identifier" unsaved-value="0">
7       <comment>the user identifier</comment>
8       <column name="id" not-null="true" unique="true" sql-type="
9         BIGINT"/>
9       <generator class="sequence"/>
10    </id>
11    <version name="version" type="int" column="object_version" not-
12      null="true"/>
12    <property name="name" type="String">
13      <comment>the user name</comment>
14      <column name="name" not-null="true" unique="false" sql-type=
15        "VARCHAR(256)"/>
15    </property>
```

```
16     <property name="email" type="String" unique='true'>
17         <comment>the user email</comment>
18         <column name="email" not-null="true" unique="false" sql-
19             type="VARCHAR(256)"/>
20     </property>
21     <property name="date" type="String">
22         <comment>the user registration date</comment>
23         <column name="date" not-null="true" unique="false" sql-type
24             ="VARCHAR(256)"/>
25     </property>
26     <property name="description" type="String">
27         <comment>the user description</comment>
28         <column name="description" not-null="true" unique="false"
29             sql-type="VARCHAR(256)"/>
30     </property>
31     <property name="status" type="Integer">
32         <comment>the user status</comment>
33         <column name="status" not-null="true" unique="false" sql-
34             type="Integer"/>
35     </property>
36 </class>
37 </hibernate-mapping>
```

The YAML description is sometimes easier to understand and write and the content could be saved in the `db/users.yaml` file.

```
1 Samples.User.Model.User:
2   type: entity
3   table: user
4   description: Record representing a user
5   hasList: true
6   indexes:
7     id:
8       id:
9         type: identifier
10        column: id
11        not-null: true
12        unique: true
13        description: the user identifier
14   fields:
15     version:
16       type: integer
17       column: object_version
18       not-null: true
19       version: true
20       unique: false
21       description:
22     name:
23       type: string
24       length: 255
25       column: name
```

```
26     not-null: true
27     unique: false
28     description: the user name
29   email:
30     type: string
31     length: 255
32     column: email
33     not-null: true
34     unique: false
35     description: the user email
36   date:
37     type: string
38     length: 255
39     column: date
40     not-null: true
41     unique: false
42     description: the user registration date
43   description:
44     type: string
45     length: 255
46     column: description
47     not-null: true
48     unique: false
49     description: the user description
50   status:
51     type: integer
52     column: status
53     not-null: true
54     unique: false
55     description: the user status
```

These XML and YAML mapping indicate that the database table `user` is represented by the `User` tagged record declared in the `Samples.User.Model` package. The table contains a `name`, `description`, `email` and a `date` column members which are a string. It also has a `status` column which is an integer. The table primary key is represented by the `id` column. The `version` column is a special column used by the optimistic locking.

4.2 Generating the Ada model and SQL schema

The Dynamo code generator is then used to generate the package and Ada records that represent our data model. The generator also generates the database SQL schema so that tables can be created easily in the database.

```
1  dynamo generate db
```

The generator will build the package specification and body for `Samples.User.Model` package. The files are created in `src/model` to make it clear that these files are model files that are generated.

The database table `user` is represented by the Ada tagged record `User_Ref`. The record members are not visible and to access the attributes it is necessary to use getter or setter operations.

The SQL files are generated for every supported database in the `db/mysql`, `db/sqlite` and `db/postgresql` directories. The generator generates two SQL files in each directory:

- A first SQL file that allows to create the tables in the database. The file name uses the pattern `create-name-driver`.
- A second SQL file that contains `DROP` statements to erase the database tables. The file name uses the pattern `drop-name-driver`.

When you modify the UML, XML or YAML model files, you should generate again the Ada and SQL files. Even though these files can be generated, it is recommended to store these generated files in a versioning systems such as `git` because this helps significantly in tracking changes in the data model.

4.3 Getting a Database Connection

To access the database, we will need a database connection. These connections are obtained from a factory and they are represented by a session object.

The session factory is the entry point to obtain a database session.

```
1 with ADO.Sessions;  
2 with ADO.Sessions.Factory;  
3 ...  
4 Factory : ADO.Sessions.Factory.Session_Factory;
```

The factory can be initialized by giving a URI string that identifies the driver and the information to connect to the database. Once created, the factory returns a session object to connect to that database. To connect to another database, another factory is necessary.

To get access to a MySQL database, the factory could be initialized as follows:

```
1 ADO.Sessions.Factory.Create (Factory, "mysql://localhost:3306/  
ado_test?user=test");
```

And to use an SQLite database, you could use:

```
1 ADO.Sessions.Factory.Create (Factory, "sqlite:tests.db");
```

For a PostgreSQL database, the factory would look like:

```
1 ADO.Sessions.Factory.Create (Factory, "postgresql://localhost:5432/  
ado_test?user=test");
```


Factory initialization is done once when an application starts. The same factory object can be used by multiple tasks.

4.4 Opening a Session

The session is created by using the `Get_Session` or the `Get_Master_Session` function of the factory. Both function return a session object associated with a database connection. The `Get_Session` will return a `Session` object which is intended to provide read-only access to the database. The `Get_Master_session` returns a `Master_Session` object which provides a read-write access to the database.

In a typical MySQL Master/Slave replication, the `Master_Session` will refer to a connection to the MySQL master while the `Session` will refer to a slave. With an SQLite database, both sessions will in fact share the same SQLite internal connection.

To load or save the user object in the database, we need a `Master_Session` database connection:

```
1 with ADO.Sessions;  
2 ...  
3   Session : ADO.Sessions.Master_Session := Factory.Get_Master_Session;
```

4.5 Creating a database record

To create our first database record, we will declare a variable that will represent the new database record. The `User_Ref` represents a reference to such record.

```
1 with Samples.User.Model;  
2 ...  
3   User : Samples.User.Model.User_Ref
```

After this declaration, the variable does not refer to any database record but we can still set some fields:

```
1   User.Set_Name ("Harry");  
2   User.Set_Age (17);
```

To save the object in the database, we just need to call the `Save` operation. To save the object, we need a database session that is capable of updating and inserting new rows. If the object does not yet have a primary key, and if the primary key allocation mode is set to `hi lo`, the ADO runtime will allocate a new unique primary key before inserting the new row.

```
1   User.Save (Session);
```

The primary key can be obtained after the first `Save` with the following operation:

```
1   Id : ADO.Identifier := User.Get_Id;
```

4.6 Loading a database record

Loading a database record is quite easy and the ADO framework proposes two mechanisms. First, let's declare our user variable:

```
1   Harry : User_Ref;
```

Then we can load the user record from the primary key identifier (assuming the identifier is "23"):

```
1   Harry.Load (Session, 23);
```

If the user cannot be found, the `Load` operation will raise the `NOT_FOUND` exception.

In many cases, we may not know the primary key but a search on one or several columns may be necessary. For this, we can create a query filter and use the `Find` operation. To use a query filter, we need first to declare a `Query` object:

```
1   with ADO.SQL;  
2   ...  
3   Query : ADO.SQL.Query;
```

On the query object, we have to define the filter which represents the condition and set the possible parameters used by the filter.

```
1   Query.Bind_Param (1, "Harry");  
2   Query.Set_Filter ("name = ?");
```

Once the query is defined and initialized, we can find the database record:

```
1   Found : Boolean;  
2   ...  
3   User.Find (Session, Query, Found);
```

Unlike the `Load` operation, `Find` does not raise an exception but instead returns a boolean value telling whether the record was found or not. The database query (and filter) has to return exactly one record to consider the object as found.

4.7 Getting a list of objects

When several records have to be read, it is necessary to use the `List` operation together with a vector object.

```
1  Users : User_Vector;
```

The `List` operation gets the vector object, the database session and the query filter. If the vector contained some elements, they are removed and replaced by the query result.

```
1  List (Users, Session, Query);
```

4.8 Running SQL queries

Sometimes it is necessary to execute SQL queries to be able to get the result without having it to be mapped to an Ada record. For this, we are going to use the `ADO.Statements`

```
1  with ADO.Statements;  
2  ...  
3  Statement : ADO.Statements.Query_Statement := Session.  
    Create_Statement ("SELECT COUNT(*) FROM user");
```

and then execute it and retrieve the result.

```
1  Statement.Execute;  
2  if not Statement.Has_Elements then  
3    Put_Line ("SQL count() failed")  
4  else  
5    Put_Line (Integer'Image (Statement.Get_Integer (0)));  
6  end if;
```

5 Session

The `ADO.Sessions` package defines the control and management of database sessions. The database session is represented by the `Session` or `Master_Session` types. It provides operation to create a database statement that can be executed. The `Session` type is used to represent read-only database sessions. It provides operations to query the database but it does not allow to update or delete content. The `Master_Session` type extends the `Session` type to provide write access and it provides operations to get update or delete statements. The differentiation between the two sessions is provided for the support of database replications with databases such as MySQL.

5.1 Connection string

The database connection string is an URI that specifies the database driver to use as well as the information for the database driver to connect to the database. The driver connection is a string of the form:

```
1 driver://[host][:port]/[database][?property1][=value1]...
```

The database connection string is passed to the session factory that maintains connections to the database (see `ADO.Sessions.Factory`).

5.2 Database Drivers

Database drivers provide operations to access the database. These operations are specific to the database type and the `ADO.Drivers` package among others provide an abstraction that allows to make the different databases look like they have almost the same interface.

A database driver exists for SQLite/SQLCipher, MySQL and PostgreSQL. The driver is either statically linked to the application or it can be loaded dynamically if it was built as a shared library. For a dynamic load, the driver shared library name must be prefixed by `libada_ado_`. For example, for a `mysql` driver, the shared library name is `libada_ado_mysql.so`.

Driver name	Database
mysql	MySQL, MariaDB
sqlite	SQLite, SQLCipher
postgresql	PostgreSQL

The database drivers are initialized automatically but in some cases, you may want to control some database driver configuration parameter. In that case, the initialization must be done only once before creating a session factory and getting a database connection. The initialization can be made using a property file which contains the configuration for the database drivers and the database connection properties. For such initialization, you will have to call one of the `Initialize` operation from the `ADO.Drivers` package.

```
1 ADO.Drivers.Initialize ("db.properties");
```

The set of configuration properties can be set programatically and passed to the `Initialize` operation.

```
1 Config : Util.Properties.Manager;  
2 ...  
3 Config.Set ("ado.database", "sqlite:///mydatabase.db");  
4 Config.Set ("ado.queries.path", ".;db");  
5 ADO.Drivers.Initialize (Config);
```

Once initialized, a configuration property can be retrieved by using the `Get_Config` operation.

```
1 URI : constant String := ADO.Drivers.Get_Config ("ado.database");
```

Dynamic loading of database drivers is disabled by default for security reasons and it can be enabled by setting the following property in the configuration file:

```
1 ado.drivers.load=true
```

Dynamic loading is triggered when a database connection string refers to a database driver which is not known.

5.2.1 MySQL Database Driver

The MySQL database driver can be initialize explicitly by using the `ado_mysql` GNAT project and calling the initialization procedure.

```
1 ADO.MySql.Initialize ("db.properties");
```

The set of configuration properties can be set programatically and passed to the `Initialize` operation.

```
1 Config : Util.Properties.Manager;  
2 ...  
3 Config.Set ("ado.database", "mysql://localhost:3306/ado_test");  
4 Config.Set ("ado.queries.path", ".;db");  
5 ADO.MySql.Initialize (Config);
```

The MySQL database driver supports the following properties:

Name	Description
user	The user name to connect to the server
password	The user password to connect to the server
socket	The optional Unix socket path for a Unix socket base connection
encoding	The encoding to be used for the connection (ex: UTF-8)

5.2.2 SQLite Database Driver

The SQLite database driver can be initialized explicitly by using the `ado_sqlite` GNAT project and calling the initialization procedure.

```
1 ADO.SQLite.Initialize ("db.properties");
```

The set of configuration properties can be set programatically and passed to the `Initialize` operation.

```
1 Config : Util.Properties.Manager;  
2 ...  
3 Config.Set ("ado.database", "sqlite:regtests.db?synchronous=OFF&  
   encoding=UTF-8");  
4 Config.Set ("ado.queries.path", ".;db");  
5 ADO.SQLite.Initialize (Config);
```

The SQLite database driver will pass all the properties as SQLite `pragma` allowing the configuration of the SQLite database. When the driver is compiled with the `-XADO_USE_SQLCIPHER=yes` option, the SQLCipher runtime will be used and you can either use a standard SQLite database or an SQLCipher database. The database encryption key must be configured as a `key` configuration property, for example:

```
1 Config.Set ("ado.database",  
2           "sqlite:regtests.cipher?key='db-password'");
```

5.2.3 PostgreSQL Database Driver

The PostgreSQL database driver can be initialize explicitly by using the `ado_postgresql` GNAT project and calling the initialization procedure.

```
1 ADO.Postgresql.Initialize ("db.properties");
```

The set of configuration properties can be set programatically and passed to the `Initialize` operation.

```
1 Config : Util.Properties.Manager;  
2 ...  
3 Config.Set ("ado.database", "postgresql://localhost:5432/ado_test?  
    user=ado&password=ado");  
4 Config.Set ("ado.queries.path", ".;db");  
5 ADO.Postgresql.Initialize (Config);
```

The PostgreSQL database driver supports the following properties:

Name	Description
user	The user name to connect to the server
password	The user password to connect to the server

5.3 Session Factory

The session factory is the entry point to obtain a database session. The `ADO.Sessions.Factory` package defines the factory for creating sessions.

```
1 with ADO.Sessions.Factory;  
2 ...  
3 Sess_Factory : ADO.Sessions.Factory;
```

The session factory can be initialized by using the `Create` operation and by giving a URI string that identifies the driver and the information to connect to the database. The session factory is created only once when the application starts.

```
1 ADO.Sessions.Factory.Create (Sess_Factory, "mysql://localhost:3306/  
    ado_test?user=test");
```

Having a session factory, one can get a database by using the `Get_Session` or `Get_Master_Session` function. Each time this operation is called, a new session is returned. The session is released when the session variable is finalized.

```
1 DB : ADO.Sessions.Session := Sess_Factory.Get_Session;
```

The session factory is also responsible for maintaining some data that is shared by all the database connections. This includes:

- the sequence generators used to allocate unique identifiers for database tables,
- the entity cache,
- some application specific global cache.

5.4 Database Caches

The ADO cache manager allows to create and maintain cache of values and use the cache from the SQL expander to replace cached values before evaluating the SQL. The SQL expander identifies constructs as follows:

```
1 $cache_name[entry-name]
```

and look for the cache identified by `cache_name` and then replace the cache entry registered with the name `entry-name`.

The cache manager is represented by the `Cache_Manager` type and the database session contains one cache manager. Applications may use their own cache in that case they will declare their cache as follows:

```
1 M : ADO.Caches.Cache_Manager;
```

A cache group is identified by a unique name and is represented by the `Cache_Type` base class. The cache group instance is registered in the cache manager by using the `Add_Cache` operation.

5.5 Sequence Generators

The sequence generator is responsible for creating unique ID's across all database objects. Each table can be associated with a sequence generator which is configured on the session factory. The sequence generators are shared by several database sessions and the implementation is thread-safe. The sequence generator provides an efficient alternative to the `AUTO INCREMENT` support provided by SQL databases.

The `HiLo_Generator` implements a simple High Low sequence generator by using sequences that avoid to access the database. The `Snowflake_Generator` implements an efficient algorithm that can be used in distributed environments.

For example, to allocate a new database identifier for a table `user`, you could write:


```
1 DB : ADO.Sessions.Master_Session := ...;
2 Id : ADO.Identifier;
3 ...
4 DB.Allocate (Name => "user", Id => Id);
```

5.5.1 HiLo Sequence Generator

The HiLo sequence generator. This sequence generator uses a database table `ado_sequence` to allocate blocks of identifiers for a given sequence name. The sequence table contains one row for each sequence. It keeps track of the next available sequence identifier (in the `value` column).

To allocate a sequence block, the HiLo generator obtains the next available sequence identified and updates it by adding the sequence block size. The HiLo sequence generator will allocate the identifiers until the block is full after which a new block will be allocated.

5.5.2 Snowflake ID Generator

The snowflake generator is a form of unique ID generator that can be used in distributed environment. It generates a 64-bit number decomposed in a first 42 bits representing a timestamp since a given epoch in milliseconds. The next N bits represent the machine ID and the last P bits a sequence number. Some snowflake generators are using 10 bits for the machine ID and 12 bits for the sequence number. Some others are using 13 bits for a shard ID and 10 bits for the sequence number. To give some flexibility, the snowflake generator is provided as a generic package that must be instantiated with two constants:

- a `Node_Bits` representing the number of bits for the machine ID, shard ID or node ID,
- a `Sequence_Bits` representing the number of bits for the sequence.

The timestamp will use $64 - \text{Node_Bits} - \text{Sequence_Bits}$ bits.

For example, the package is first instantiated with the desired configuration:

```
1 package My_Snowflake is
2   new ADO.Sequences.Snowflake (Node_Bits => 12,
3                               Sequence_Bits => 10);
```

An instance of the generator must be created and registered in the session factory. The generator is assigned a name (in most cases a database table name) which will be used to identify it by the `Allocate` procedure. The Snowflake generator needs an epoch timestamp representing the start of epoch for the sequence generation and a machine ID or node ID which should be unique and specific to each server instance that will create and use such generator.

```
1  S : ADO.Sequences.Generator_Access
2      := My_Snowflake.Create_Snowflake_Generator
3          (Sess_Factory'Unchecked_Access, "user", Epoch, 10);
4  begin
5      Sess_Factory.Set_Generator (S);
```

6 Database Statements

The `ADO.Statements` package provides high level operations to construct database statements and execute them. They allow to represent SQL statements (prepared or not) and provide support to execute them and retrieve their result. The SQL statements are represented by several Ada types depending on their behavior:

- The `Statement` type represents the base type for all the SQL statements.
- The `Query_Statement` type is intended to be used for database query statements and provides additional operations to retrieve results.
- The `Update_Statement` type targets the database update statements and it provides specific operations to update fields. The `Insert_Statement` extends the `Update_Statement` type and is intended for database insertion.
- The `Delete_Statement` type is intended to be used to remove elements from the database.

The database statements are created by using the database session and by providing the SQL or the named query to be used.

6.1 Query Parameters

Query parameters are represented by the `Parameter` type which can represent almost all database types including boolean, numbers, strings, dates and blob. Parameters are put in a list represented by the `Abstract_List` or `List` types.

A parameter is added by using either the `Bind_Param` or the `Add_Param` operation. The `Bind_Param` operation allows to specify either the parameter name or its position. The `Add_Param` operation adds the parameter at end of the list and uses the last position. In most cases, it is easier to bind a parameter with a name as follows:

```
1 Query.Bind_Param ("name", "Joe");
```

and the SQL can use the following construct:

```
1 SELECT * FROM user WHERE name = :name
```

When the `Add_Param` is used, the parameter is not associated with any name but it as a position index. Setting a parameter is easier:

```
1 Query.Add_Param ("Joe");
```

but the SQL cannot make any reference to names and must use the `?` construct:

```
1 SELECT * FROM user WHERE name = ?
```

6.1.1 Parameter Expander

The parameter expander is a mechanism that allows to replace or inject values in the SQL query by looking at an operation provided by the Expander interface. Such expander is useful to replace parameters that are global to a session or to an application.

6.2 Query Statements

The database query statement is represented by the `Query_Statement` type. The `Create_Statement` operation is provided on the `Session` type and it gets the SQL to execute as parameter. For example:

```
1 Stmt : ADO.Statements.Query_Statement := Session.Create_Statement
2      ("SELECT * FROM user WHERE name = :name");
```

After the creation of the query statement, the parameters for the SQL query are provided by using either the `Bind_Param` or `Add_Param` procedures as follows:

```
1 Stmt.Bind_Param ("name", name);
```

Once all the parameters are defined, the query statement is executed by calling the `Execute` procedure:

```
1 Stmt.Execute;
```

Several operations are provided to retrieve the result. First, the `Has_Elements` function will indicate whether some database rows are available in the result. It is then possible to retrieve each row and proceed to the next one by calling the `Next` procedure. The number of rows is also returned by the `Get_Row_Count` function. A simple loop to iterate over the query result looks like:

```
1 while Stmt.Has_Elements loop
2   Id := Stmt.Get_Identifier (1);
3   ...
4   Stmt.Next;
5 end loop;
```

6.3 Named Queries

Ada Database Objects provides a small framework which helps in using complex SQL queries in an application by using named queries. The benefit of the framework are the following:

- The SQL query result are directly mapped in Ada records,

- It is easy to change or tune an SQL query without re-building the application,
- The SQL query can be easily tuned for a given database.

The database query framework uses an XML query file:

- The XML query file defines a mapping that represents the result of SQL queries,
- The XML mapping is used by Dynamo code generator to generate an Ada record,
- The XML query file also defines a set of SQL queries, each query being identified by a unique name,
- The XML query file is read by the application to obtain the SQL query associated with a query name,
- The application uses the `List` procedure generated by Dynamo.

6.3.1 XML Query File

The XML query file uses the `query-mapping` root element. It should define at most one `class` mapping and several `query` definitions. The `class` definition should come first before any `query` definition.

```
1 <query-mapping>
2   <class>...</class>
3   <query>...</query>
4 </query-mapping>
```

6.3.2 SQL Result Mapping

The XML query mapping is very close to the database XML table mapping. The difference is that there is no need to specify any table name nor any SQL type. The XML query mapping is used to build an Ada record that correspond to query results. Unlike the database table mapping, the Ada record will not be tagged and its definition will expose all the record members directly.

The following XML query mapping:

```
1 <query-mapping>
2   <class name='Samples.Model.User_Info'>
3     <property name="name" type="String">
4       <comment>the user name</comment>
5     </property>
6     <property name="email" type="String">
7       <comment>the email address</comment>
8     </property>
9   </class>
10 </query-mapping>
```

will generate the following Ada record and it will instantiate the Ada container `Vectors` generic to provide a support for vectors of the record:

```
1 package Samples.Model is
2   type User_Info is record
3     Name : Unbounded_String;
4     Email : Unbounded_String;
5   end record;
6   package User_Info_Vectors is
7     new Ada.Containers.Vectors (Index_Type => Natural,
8                                Element_Type => User_Info,
9                                "=" => "=");
10    subtype User_Info_Vector is User_Info_Vectors.Vector;
11 end Samples.Model;
```

A `List` operation is also generated and can be used to execute an SQL query and have the result mapped in the record.

The same query mapping can be used by different queries.

After writing or updating a query mapping, it is necessary to launch the Dynamo code generator to generate the corresponding Ada model.

6.3.3 SQL Queries

The XML query file defines a list of SQL queries that the application can use. Each query is associated with a unique name. The application will use that name to identify the SQL query to execute. For each query, the file also describes the SQL query pattern that must be used for the query execution.

```
1 <query-mapping>
2   <query name='user-list' class='Samples.Model.User_Info'>
3     <sql driver='mysql'>
4       SELECT u.name, u.email FROM user AS u
5     </sql>
6     <sql driver='sqlite'>
7       ...
8     </sql>
9     <sql-count driver='mysql'>
10      SELECT COUNT(*) FROM user AS u
11    </sql-count>
12  </query>
13 </query-mapping>
```

The query contains basically two SQL patterns. The `sql` element represents the main SQL pattern. This is the SQL that is used by the `List` operation. In some cases, the result set returned by the query is limited to return only a maximum number of rows. This is often use in paginated lists.

The `sql-count` element represents an SQL query to indicate the total number of elements if the SQL query was not limited.

The `sql` and `sql-count` XML element can have an optional `driver` attribute. When defined, the attribute indicates the database driver name that is specific to the query. When empty or not defined, the SQL is not specific to a database driver.

For each query, the Dynamo code generator generates a query definition instance which can be used in the Ada code to be able to use the query. Such instance is static and readonly and serves as a reference when using the query. For the above query, the Dynamo code generator generates:

```
1 package Samples.User.Model is
2   Query_User_List : constant ADO.Queries.Query_Definition_Access;
3 private
4   ...
5 end Samples.User.Model;
```

When a new query is added, the Dynamo code generator must be launched to update the generated Ada code.

6.3.4 Using Named Queries

In order to use a named query, it is necessary to create a query context instance and initialize it. The query context holds the information about the query definition as well as the parameters to execute the query. It provides a `Set_Query` and `Set_Count_Query` operation that allows to configure the named query to be executed. It also provides all the `Bind_Param` and `Add_Param` operations to allow giving the query parameters.

```
1 with ADO.Sessions;
2 with ADO.Queries;
3 ...
4 Session : ADO.Sessions.Session := Factory.Get_Session;
5 Query    : ADO.Queries.Context;
6 Users    : Samples.User.Model.User_Info_Vector;
7 ...
8 Query.Set_Query (Samples.User.Model.Query_User_List);
9 Samples.User.Model.List (Users, Session, Query);
```

To use the `sql-count` part of the query, you will use the `Set_Count_Query` with the same query definition. You will then create a query statement from the named query context and run the query. Since the query is expected to contain exactly one row, you can use the `Get_Result_Integer` to get the first row and column result. For example:

```
1 Query.Set_Count_Query (Samples.User.Model.Query_User_List);
2 ...
```

```
3 Stmt : ADO.Statements.Query_Statement
4   := Session.Create_Statement (Query);
5 ...
6 Stmt.Execute;
7 ...
8 Count : Natural := Stmt.Get_Result_Integer;
```

You may also use the `ADO.Datasets.Get_Count` operation which simplifies these steps in:

```
1 Query.Set_Count_Query (Samples.User.Model.Query_User_List);
2 ...
3 Count : Natural := ADO.Datasets.Get_Count (Session, Query);
```


7 Model Mapping

A big benefit when using ADO is the model mapping with the Ada and SQL code generator.

The model describes the database tables, their columns and relations with each others. It is then used to generate the Ada implementation which provides operations to create, update and delete records from the database and map them in Ada transparently.

The model can be defined in:

- UML with a modeling tool that exports the model in XMI,
- XML files following the Hibernate description,
- YAML files according to the Doctrine mapping.

This chapter focuses on the YAML description.

7.1 Table definition

In YAML, the type definition follows the pattern below:

```
1 <table-type-name>:  
2   type: entity  
3   table: <table-name>  
4   description: <description>  
5   hasList: true|false  
6   indexes:  
7   id:  
8   fields:  
9   oneToOne:  
10  oneToMany:
```

The *table-type-name* represents the Ada type name with the full package specification. The code generator will add the *_Ref* prefix to the Ada type name to define the final type with reference counting. A private type is also generated with the *_Impl* prefix.

The YAML fields have the following meanings:

Field	Description
type	Must be 'entity' to describe a database table
table	The name of the database table. This must be a valid SQL name
description	A comment description for the table and type definition
hasList	When true , a List operation is also generated for the type

Field	Description
indexes	Defines the indexes for the table
id	Defines the primary keys for the table
fields	Defines the simple columns for the table
oneToOne	Defines the one to one table relations
oneToMany	Defines the one to many table relations

7.2 Column mapping

Simple columns are represented within the `fields` section.

```

1 <table-type-name>:
2   fields:
3     <member-name>:
4       type: <type>
5       length: <length>
6       description: <description>
7       column: <column-name>
8       not-null: true|false
9       unique: true|false
10      readonly: true|false
11      version: false

```

The YAML fields have the following meanings:

Field	Description
type	The column type. This type maps to an Ada type and an SQL type
length	For variable length columns, this is the maximum length of the column
description	A comment description for the table and type definition
column	The database table column name
not-null	When true, indicates that the column cannot be null
unique	When true, indicates that the column must be unique in the table rows
readonly	When true, the column cannot be updated. The <code>Save</code> operation will ignore updated.
version	Must be 'false' for simple columns

The `type` column describes the type of the column using a string that is agnostic of the Ada and SQL languages. The mapping of the type to SQL depends on the database. The `not-null` definition has an impact on the Ada type since when the column can be null, a special Ada type is required to represent that null value.

The `ADO.Nullable_X` types are all represented by the following record:

```
1  type Nullable_X is record
2    Value    : X := <default-value>;
3    Is_Null  : Boolean := True;
4  end record;
```

The `Is_Null` boolean member must be checked to see if the value is null or not. The comparison operation (`=`) ignores the `Value` comparison when one of the record to compare has `Is_Null` set.

Type	not-null	SQL	Ada
boolean	true	TINYINT	Boolean
	false	TINYINT	ADO.Nullable_Boolean
byte	true	TINYINT	-
	false	TINYINT	-
integer	true	INTEGER	Integer
	false	INTEGER	ADO.Nullable_Integer
long	true	BIGINT	Long_Long_Integer
	false	BIGINT	ADO.Nullable_Long_Integer
identifier		BIGINT	ADO.Identifier
entity_type	true	INTEGER	ADO.Entity_Type
	false	INTEGER	ADO.Nullable_Entity_Type
string	true	VARCHAR(N)	Unbounded_String
	false	VARCHAR(N)	ADO.Nullable_String
date	true	DATE	Ada.Calendar.Time
	false	DATE	ADO.Nullable_Time
time	true	DATETIME	Ada.Calendar.Time
	false	DATETIME	ADO.Nullable_Time
blob		BLOB	ADO.Blob_Ref

The `identifier` type is used to represent a foreign key mapped to a `BIGINT` in the database. It is always represented by the Ada type `ADO.Identifier` and the null value is represented by the special value `ADO.NO_IDENTIFIER`.

The `blob` type is represented by an Ada stream array held by a reference counted object. The reference can be null.

The `entity_type` type allows to uniquely identify the type of a database entity. Each database table is associated with an `entity_type` unique value. Such value is created statically when the database schema is created and populated in the database. The `entity_type` values are maintained in the `entity_type` ADO database table.

7.3 Primary keys

Primary keys are used to uniquely identify a row within a table. For the ADO framework, only the identifier and string primary types are supported.

```

1 <table-type-name>:
2   id:
3     <member-name>:
4       type: {identifier|string}
5       length: <length>
6       description: <description>
7       column: <column-name>
8       not-null: true
9       unique: true
10      version: false
11      generator:
12        strategy: {none|auto|sequence}

```

The `generator` section describes how the primary key is generated.

Strategy	description
none	the primary key is managed by the application
auto	use the database auto increment support
sequence	use the ADO sequence generator

7.4 Relations

A one to many relation is described by the following YAML description:

```
1 <table-type-name>:  
2   oneToMany:  
3     <member-name>:  
4       type: <model-type>  
5       description: <description>  
6       column: <column-name>  
7       not-null: true|false  
8       readonly: true|false
```

This represents the foreign key and this YAML description is to be put in the table that holds it.

The `type` definition describes the type of object at the end of the relation. This can be the `identifier` type which means the relation will not be strongly typed and mapped to the `ADO.Identifier` type. But it can be the table type name used for another table definition. In that case, the code generator will generate a getter and setter that will use the object reference instance.

Circular dependencies are allowed within the same Ada package. That is, two tables can reference each other as long as they are defined in the same Ada package. A relation can use a reference of a type declared in another YAML description from another Ada package. In that case, `with` clauses are generated to import them.

7.5 Versions

Optimistic locking is a mechanism that allows updating the same database record from several transactions without having to take a strong row lock that would block transactions. By having a version column that is incremented after each change, it is possible to detect that the database row was modified when we want to update it. When this happens, the optimistic lock exception `ADO.Objects.LAZY_LOCK` is raised and it is the responsibility of the application to handle the failure by retrying the update.

For the optimistic locking to work, a special integer based column must be declared.

```
1 <table-type-name>:  
2   fields:  
3     <member-name>:  
4       type: <type>  
5       description: <description>  
6       column: <column-name>  
7       not-null: true  
8       unique: false  
9       version: true
```

The generated Ada code gives access to the version value but it does not allow its modification. The version column is incremented only by the `Save` procedure and only if at least one field of the record

was modified (otherwise the `Save` has no effect). The version number starts with the value 1. ## Objects When a database table is mapped into an Ada object, the application holds a reference to that object through the `Object_Ref` type. The `Object_Ref` tagged type is the root type of any database record reference. Reference counting is used so that the object can be stored, shared and the memory management is handled automatically. It defines generic operations to be able to:

- load the database record and map it to the Ada object,
- save the Ada object into the database either by inserting or updating it,
- delete the database record.

The Dynamo code generator will generate a specific tagged type for each database table that is mapped. These tagged type will inherit from the `Object_Ref` and will implement the required abstract operations. For each of them, the code generator will generate the `Get_X` and `Set_X` operation for each column mapped in Ada.

Before the `Object_Ref` is a reference, it does not hold the database record itself. The `ADO.Objects.Object_Record` tagged record is used for that and it defines the root type for the model representation. The type provides operations to modify a data field of the record while tracking its changes so that when the `Save` operation is called, only the data fields that have been modified are updated in the database. An application will not use nor access the `Object_Record`. The Dynamo code generator generates a private type to make sure it is only accessed through the reference.

Several predicate operations are available to help applications check the validity of an object reference:

Function	Description
<code>Is_Null</code>	When returning True, it indicates the reference is NULL.
<code>Is_Loaded</code>	When returning True, it indicates the object was loaded from the database.
<code>Is_Inserted</code>	When returning True, it indicates the object was inserted in the database.
<code>Is_Modified</code>	When returning True, it indicates the object was modified and must be saved.

Let's assume we have a `User_Ref` mapped record, an instance of the reference would be declared as follows:

```
1 with Samples.User.Model;  
2 ...  
3 User : Samples.User.Model.User_Ref;
```

After this declaration, the reference is null and the following assumption is true:

```
1 User.Is_Null and not User.Is_Loaded and not User.Is_Inserted
```

If we set a data field such as the name, an object is allocated and the reference is no longer null.

```
1 User.Set_Name ("Ada Lovelace");
```

After this statement, the following assumption is true:

```
1 not User.Is_Null and not User.Is_Loaded and not User.Is_Inserted
```

With this, it is therefore possible to identify that this object is not yet saved in the database. After calling the `Save` procedure, a primary key is allocated and the following assumption becomes true:

```
1 not User.Is_Null and not User.Is_Loaded and User.Is_Inserted
```

7.6 Loading Objects

Three operations are generated by the Dynamo code generator to help in loading a object from the database: two `Load` procedures and a `Find` procedure. The `Load` procedures are able to load an object by using its primary key. Two forms of `Load` are provided: one that raises the `ADO.Objects.NOT_FOUND` exception and another that returns an additional `Found` boolean parameter. Within the application, if the database row is expected to exist, the first form should be used. In other cases, when the application expects that the database record may not exist, the second form is easier and avoids raising and handling an exception for a common case.

```
1 User.Load (Session, 1234);
```

The `Find` procedure allows to retrieve a database record by specifying a filter. The filter object is represented by the `ADO.SQL.Query` tagged record. A simple query filter is declared as follows:

```
1 Filter : ADO.SQL.Query;
```

The filter is an SQL fragment that is inserted within the `WHERE` clause to find the object record. The filter can use parameters that are configured by using the `Bind_Param` or `Add_Param` operations. For example, to find a user from its name, the following filter could be set:

```
1 Filter.Set_Filter ("name = :name");  
2 Filter.Bind_Param ("name", "Ada Lovelace");
```

Once the query filter is initialized and configured with its parameters, the `Find` procedure can be called:

```
1 Found : Boolean;  
2 ...
```

```
3 User.Find (Session, Filter, Found);
```

The `Find` procedure does not raise an exception if the database record is not found. Instead, it returns a boolean status in the `Found` output parameter. The `Find` procedure will execute an SQL `SELECT` statement with a `WHERE` clause to retrieve the database record. The `Found` output parameter is set when the query returns exactly one row.

7.7 Modifying Objects

To modify an object, applications will use one of the `Set_X` operation generated for each mapped column. The ADO runtime will keep track of which data fields are modified. The `Save` procedure must be called to update the database record. When calling it, an SQL `UPDATE` statement is generated to update the modified data fields.

```
1 User.Set_Status (1);  
2 User.Save (Session);
```

7.8 Deleting Objects

Deleting objects is made by using the `Delete` operation.

```
1 User.Delete (Session);
```

Sometimes you may want to delete an object without having to load it first. This is possible by delete an object without loading it. For this, set the primary key on the object and call the `Delete` operation:

```
1 User.Set_Id (42);  
2 User.Delete (Session);
```


8 Database schema migration

In the lifetime of an application, the database schema may be changed and these changes must be applied to the database when a new version of the application is installed. Each application can implement its own mechanism to track the database schema change and apply the necessary change to the database when the application is upgraded.

8.1 Migration mechanism

An automatic tracking and update is too difficult for this library because such database migration is specific to each application. However, the Ada Database Objects library provides a mechanism to help in updating the database schema. The proposed mechanism is based on:

- the `ado_version` database table which tracks for each application module, a schema version. The schema version is changed by the developer when a new database schema is made for the module and some SQL migration script must be executed. The version is a simple integer that is incremented by the developer when the schema is modified.
- a set of SQL migration scripts that control the migration of a module from a version to the next one. The SQL migration scripts are written manually by the developer (or by some generation tool if they exist) when the database schema is modified.

The SQL migration scripts must be organized on the file system so that the library can find them and decide whether they must be executed or not. Each SQL migration script of a module must be stored in a directory with the name of that module. Then, each upgrade for a schema version of the module must also be stored in a directory with the name of the version. To control the order of execution of SQL migration scripts, each script file must be prefixed by a unique order. SQL scripts are executed in increasing order. Following the unique order, a tag separated by `-` indicates the database driver that corresponds to the SQL script. The special tag `all` indicates that the script must be executed for all database drivers. This allows to have specific SQL for different database types such as SQLite, PostgreSQL or MySQL.

To summarize, the layout is the following:

```
1 db/migrate/<module>/<version>/depend.conf
2 db/migrate/<module>/<version>/<order>-all-<name>.sql
3 db/migrate/<module>/<version>/<order>-<driver>-<name>.sql
```

The database migration support is decomposed in several steps each represented by a separate procedure so that it provides a fine grain control over the migration execution. For example it is possible to print the list of SQL scripts without executing them. It is also possible to perform partial migration and upgrade only a subset of the list by stopping at any time.

To proceed with the migration, the library will do the following:

- it scans the `db/migrate` directory which contains the module SQL migration scripts. For each database module, it identifies the module versions that must be upgraded. A module needs an upgrade when there is a `<version>` directory that exists and is higher than the current schema version recorded in `ado_version`. This step is done by the `ADO.Schemas.Databases.Scan_Migration` procedure. Upon completion of this step, we get a list of `Update_Type` records that indicate every upgrade that must be executed.
- the list of upgrades must then be sorted to honor the dependencies between the database modules. This step must be done to make sure that the dependent schemas are updated to the specified dependent version. This step is executed by the `ADO.Schemas.Databases.Sort_Migration` procedure.
- having a sorted list of upgrade, it is now possible to scan the directory that contains the SQL scripts (ie `db/migrate/<module>/<version>`). When reading that directory, we only look at the SQL files with the `.sql` extension. The file name must follow the pattern `<order>-<tag>-<title>.sql` (or the regular expression `[0-9]+-(all|mysql|postgresql|sqlite)-.*\.sql`). The SQL file is taken into account if it uses the tag `all` or uses the database driver name of the database being upgraded. The final list of SQL files is then sorted. This step is handled by the `ADO.Schemas.Prepare_Migration` procedure. It produces a list of files with their absolute path in the order in which they must be executed for the module upgrade for the specific version.
- the last step is to execute the SQL statements from the list of files computed previously. During this step, the SQL file is read and split into SQL statements that are executed on the database. Once every statement is executed, the next file is processed until the last one. At the end, the version associated with the database module is update to the upgraded version. If the `ado_version` table does not contain the module, an entry is created with the version. This step is made by the `ADO.Schemas.Run_Migration` procedure. The `Run_Migration` procedure will also automatically call `Prepare_Migration` to collect the list of files if necessary.

The following code extract illustrates a complete database migration support:

```
1 with ADO.Sessions;  
2 with ADO.Schemas.Databases;  
3 ...  
4 Session : ADO.Sessions.Master_Session;  
5 List    : ADO.Schemas.Databases.Upgrade_List;  
6 Files   : Util.Strings.Vectors.Vector;  
7 ...  
8     ADO.Schemas.Databases.Scan_Migration (Session, "db/migrate", List);  
9     ADO.Schemas.Databases.Sort_Migration (List);  
10    for Upgrade of List loop  
11        ADO.Schemas.Databases.Run_Migration
```

```
12         (Session, Upgrade, Files, ADO.Sessions.Execute'Access);  
13     end loop;
```

8.2 Taking into account the migration during development

This section gives some tips to take into account database schema changes during development.

When there is a database schema change in a module, the version should be incremented and one or several SQL scripts to upgrade only to the new version are written. The schema change must be identified and analyzed either manually or with some tool. For example if a new column is added to some table, you will have to write some SQL that looks like:

```
1 ALTER TABLE awa_post ADD COLUMN `read_count` INTEGER NOT NULL;
```

When a schema change is made on the module and that module depends on another one, it is necessary for correct dependency order to write the `depend.conf` file that indicates the list of modules and their version we are depending on.

For example, if we update a module named `blogs` and that module depends on `awa` which itself depends on `ado`, the `depend.conf` file must specify these two modules with their specific versions. For example:

```
1 awa:1 ado:2
```

It is also possible to create intermediate upgrade versions for the same module. This is useful when there are incremental development steps but also when some migration becomes too complex. By splitting the complex migration in simpler and small incremental steps, this will help when the migration must be executed because it will be possible to execute one simple step at a time.

9 Troubleshooting

9.1 Change the log configuration

The ADO runtime uses the logging framework provided by Ada Utility Library. By default, logging messages are disabled and the logging framework has a negligible impact on performance (less than 1 us per log).

You can customize the logging framework so that you activate logs according to your needs. In the full mode, the ADO runtime will report the SQL statements which are executed.

To control the logging, add or update the following definitions in a property file:

```
1 log4j.rootCategory=DEBUG,console,result
2
3 log4j.appender.console=Console
4 log4j.appender.console.level=WARN
5 log4j.appender.console.layout=level-message
6
7 log4j.appender.result=File
8 log4j.appender.result.File=test.log
9
10 # Logger configuration
11 log4j.logger.ADO=INFO,result
12 log4j.logger.ADO.Sessions=WARN
13 log4j.logger.ADO.Statements=DEBUG
```

The logging framework is configured by using the `Util.Log.Logging.Initialize` operation:

```
1 Util.Log.Loggers.Initialize ("config.properties");
```

which can be executed from any place (but the best place is during the application start).

You can also configure the logger in Ada by using the following code:

```
1 with Util.Properties;
2 ...
3 Log_Config : Util.Properties.Manager;
4 ...
5 Log_Config.Set ("log4j.rootCategory", "DEBUG,console");
6 Log_Config.Set ("log4j.appender.console", "Console");
7 Log_Config.Set ("log4j.appender.console.level", "ERROR");
8 Log_Config.Set ("log4j.appender.console.layout", "level-message");
9 Log_Config.Set ("log4j.logger.Util", "FATAL");
10 Log_Config.Set ("log4j.logger.ADO", "ERROR");
11 Log_Config.Set ("log4j.logger.ADO.Statements", "DEBUG");
12 Util.Log.Loggers.Initialize (Log_Config);
```

The ADO runtime has several loggers, each of them can be activated separately. The following loggers are interesting:

Logger name	Description
ADO.Drivers	Database drivers and connection to servers
ADO.Sessions	Database session management
ADO.Statements	SQL statements execution
ADO.Queries	Named queries identification and retrieval

9.2 Handling exceptions

Some exceptions are raised when there is a serious problem. The problem could be of different nature:

- there is a database connection issue,
- there is an SQL error,
- there is a data inconsistency.

The `ADO.Sessions.Connection_Error` exception is raised when the connection string used to access the database is incorrect. The connection string could be improperly formatted, a database driver may not be found, the database server may not be reachable.

The `ADO.Sessions.Session_Error` exception is raised when the `Session` object is used while it is not initialized or the connection was closed programmatically.

The `ADO.Queries.Query_Error` exception is raised when a named query cannot be found. In that case, the SQL that corresponds to the query cannot be executed.

The `ADO.Statements.SQL_Error` exception is raised when the execution of an SQL query fails. This is an indication that the SQL statement is invalid and was rejected by the database.

The `ADO.Statements.Invalid_Column` exception is raised after the execution of an SQL query when the application tries to access the result. It is raised when the program tries to retrieve a column value that does not exist.

The `ADO.Statements.Invalid_Type` exception is also raised after the execution of an SQL query when the value of a column cannot be converted to the Ada type. It occurs if a column contains a string while the application tries to get the column as an integer or date. Similarly, if a column is null and the returned Ada type does not support the nullable concept, this exception will be raised.

The `ADO.Statements.Invalid_Statement` exception is raised when you try to use and execute a `Statement` object which is not initialized.

The object layer provided by ADO raises specific exceptions.

The `ADO.Objects.NOT_FOUND` exception is raised by the generated `Load` procedure when an object cannot be found in the database.

The `ADO.Objects.INSERT_ERROR` exception is raised by the generated `Save` procedure executed the SQL INSERT statement and its execution failed.

The `ADO.Objects.UPDATE_ERROR` exception is raised by the generated `Save` procedure executed the SQL UPDATE statement and its execution failed.

The `ADO.Objects.LAZY_LOCK` exception is raised by the generated `Save` procedure executed the SQL UPDATE statement failed and the version of the object was changed.